Appendix C - Modelling steps: EViews program file

1) Create workfile (named 'ESSFC'), using annual data from 1996 to 2030:

```
wfcreate(wf = ESSFC) a 1996 2030
```

2) Upload / import time series (marked by subscript 'ts') from Excel sheet:

```
read(b2, s=sub_sheet_name) "C:\...\Excel_sheet_name.xls" yd_h_ts cons_h_ts nw_h_ts
...
```

3) Create and label model series:

```
series cons_h
cons_h.label(d) Household consumption
series yd_h
yd_h.label(d) Household disposable income
series nw_h
nw_h.label Households net wealth
...
```

4) Set sample size (entire workfile range):

```
smpl 1996 2030
```

5) Define the set of parameters to be estimated, e.g. p(1), p(2), ..., p(400):

```
coef(400) p
```

6) Estimate parameter values: simple OLS estimation equation by equation:

```
equation eq1.ls(cov=white) cons_h_ts = p(1)*yd_h_ts(-1) + p(2)*nw_h_ts(-1)
...
```

Note: White standard errors are used. Variables can be transformed in the usual way to deal with non-stationarity issues (e.g. d(log(cons_h_ts))), etc.

7) Select starting values for stocks and lagged (endogenous) variables:

```
l_h = l_h_ts                                         'Loans to households
delta_f = @mean((inv_f_ts/inv_tot_ts), "1997 2016")   'Firms investment as % of total investment
...
```

Note: the ratio of firms' investment to total investment is defined as the average value during 1997-2016, while the initial value of the stock of loans to households is set at its historical level.

8) Define fine-tuned parameters and exogenous variables:

```
r_d = 0 'interest rate on bank deposits and cash
...
```

9) Create the model (named 'ESSFC'):

```
model ESSFC
```

10) Set up system of difference equations:

```
ESSFC.append cons_h = p(1)*yd_h(-1) + p(2)*nw_h(-1) 'Household consumption (stochastic
                                                    'equation)
```

```
ESSFC.append @identity yd_h = gdp_h + wb - tau_h + int_h + t_h + ann 'Household disposable
                                                                    'income (identity)
...
```

Note: in the consumption equation, p(1) and p(2) take automatically the values estimated at point 6.

10.tris) Some series can be defined as moving averages:

```
ESSFC.append omega = @recode(@date<@dateval("2017"),@movavc(wb_ts/gdp_ts,3), @mean(wb_ts/gdp_ts,
"2015 2016"))  'Share of net wages to GDP
...
```

Note: in the example above, the share of net wages to GDP is calculated as a three-year moving average up until 2016. Starting from 2017, the average value during 2015-2016 is taken.

11) Select the baseline Scenario:

```
ESSFC.scenario "baseline"
```

12) Define the sample:

```
smpl 1998 2030
```

Note: the sample includes forecast values after 2016.

13) Create "Add factors" to improve in-sample forecast:

```
ESSFC.addassign(v) @stochastic 'or @all
ESSFC.addinit(v=n) @stochastic
```

Note: addassign(v) = create ‘Add factors’ as variable shift (as opposed to intercept shift); addinit(v=n) = initialise ‘Add factors’ in such a way that there is no residual left (several options are available).

14) Solve the model:

```
ESSFC.solve(i=a, s=d, d=d)
```

Note: i=a sets initial solution values equal to actual values in period prior to start of solution period; s=d deterministic solution (as opposed to stochastic solution); d=d means dyamic solution (as opposed to static).

15) As usual alternative scenarios / shocks to model exogenous variables can be created. For instance, a permanent cut in government consumption (-1%) in 2017 can be obtained using the code below:

```
ESSFC.scenario "scenario 1"
ESSFC.override parag
copy parag parag_1
smpl 2017 @last
parag_1 = -gdp*0.01
smpl 1998 2030
ESSFC.solve
```

Note: ‘parag’ is a parameter defining government’s autonomous consumption.